




## Adafruit IO Basics: Digital Output

Created by Todd Treece




Last updated on 2017-09-12 03:10:33 PM UTC

## Guide Contents

Guide Contents	2
Overview	3
Adafruit IO Setup	5
Creating the Digital Feed	6
Adding the Toggle Block	6
Wiring	8
Arduino Setup	10
Network Config	11
WiFi Config	11
FONA Config	12
Ethernet Config	12
Code	14

# Overview



This guide is part of a series of guides that cover the basics of using Adafruit IO. It will show you how to turn a LED on and off from Adafruit IO using any modern web browser.

If you haven't worked your way through the Adafruit IO feed and dashboard basics guides, you should do that before continuing with this guide so you have a basic understanding of Adafruit IO.

- [Adafruit IO Basics: Feeds](#)
- [Adafruit IO Basics: Dashboards](#)

You should go through the setup guides associated with your selected set of hardware, and make sure you have internet connectivity with the device before continuing. The following links will take you to the guides for your selected platform.

- [Adafruit Feather HUZZAH ESP8266 Setup Guide](#)


If you have went through all of the prerequisites for your selected hardware, you are now ready to move on to the Adafruit IO setup steps that are common between all of the hardware choices for this project. Let's get started!




# Adafruit IO Setup

The first thing you will need to do is to login to [Adafruit IO](#) and visit the **Settings** page.

Click the **VIEW AIO KEY** button to retrieve your key.



A window will pop up with your Adafruit IO. Keep a copy of this in a safe place. We'll need it later.



## Creating the Digital Feed

Next, you will need to create a feed called **Digital**. If you need help getting started with creating feeds on Adafruit IO, check out the [Adafruit IO Feed Basics guide](http://adafru.it/ioA) (<http://adafru.it/ioA>).


A screenshot of the 'Create a new Feed' form in Adafruit IO. The form has a white background with a blue 'X' close button in the top right. The title 'Create a new Feed' is at the top left. Below it, there are two input fields: 'Name' and 'Description'. The 'Name' field is a text box with a blue border and a blue background, containing the text 'Digital'. The 'Description' field is a larger text box with a gray border. At the bottom right, there are two buttons: 'Cancel' and 'Create'.

## Adding the Toggle Block

Next, add a new Toggle Block to a new or existing dashboard. Name the block whatever you would like, and set **On Text to a value of 1** and **Off Text to a value of 0**. Make sure

you have selected the **Digital** feed as the data source for the toggle.

If you need help getting started with Dashboards on Adafruit IO, check out the [Adafruit IO Dashboard Basics guide](http://adafru.it/f5m) (<http://adafru.it/f5m>).



The screenshot shows the 'Block settings' interface. At the top, it says 'Block settings' with a close button (X). Below this is a paragraph: 'In this final step, you can give your block a title and see a preview of how it will look. Customize the look and feel of your block with the remaining settings. When you are ready, click the "Create Block" button to send it to your dashboard.'


On the left, there are three input fields:

- 'Block Title' with the value 'LED'.
- 'Button On Text' with the value '1'.
- 'Button Off Text' with the value '0'.

On the right, there is a 'Block Preview' area. It shows a dark background with the title 'LED' at the top. Below the title is a toggle switch. The left half of the switch is white, and the right half is red with a white '0' in the center.

At the bottom right of the form, there are two buttons: '< Previous step' (disabled) and 'Create block' (active).

When you are finished editing the form, click *Create Block* to add the new block to the dashboard.



Next, we will look at wiring the circuit.

# Wiring


You will need the following parts for this tutorial:

- **1x** Adafruit IO compatible Feather
- **2x** jumper wires
- **1x** 560 ohm resistor
- **1x** 10mm LED

You will need to connect the following pins to the LED and resistor:

- Feather **GND** to LED **cathode** (short leg)
- Feather **Pin 5** to one leg of the **560 ohm resistor**
- LED **anode** (long leg) to the second leg of the **560 ohm resistor**

**Note:** Resistors are *not* polarized, so the 560 ohm resistor can be connected to the circuit in either direction.



fritzing




Next, let's look at the example sketch we will be using.

# Arduino Setup


You should go through the setup guides associated with your selected set of hardware, and make sure you have internet connectivity with the device before continuing. The following links will take you to the guides for your selected platform.

- [Adafruit Feather HUZAH ESP8266 Setup Guide](#)

You will need to make sure you have at least **version 2.3.1** of the Adafruit IO Arduino library installed before continuing.



For this example you will need to open the **adafruitio\_07\_digital\_out** example in the **Adafruit IO Arduino** library.



Next, we will look at the network configuration options in the sketch.

# Network Config

To configure the network settings, click on the **config.h** tab in the sketch. You will need to set your Adafruit IO username in the **IO\_USERNAME** define, and your Adafruit IO key in the **IO\_KEY** define.

```

config.h | Arduino 1.8.1

set your username and key

config.h $

/***** Adafruit IO Config *****/

// visit io.adafruit.com you need to create an account,
// or if you need your Adafruit IO key.
#define IO_USERNAME "uniontownlabs"
#define IO_KEY " "
  
```

## WiFi Config

WiFi is enabled by default in **config.h** so if you are using one of the supported WiFi boards, you will only need to modify the **WIFI\_SSID** and **WIFI\_PASS** options in the **config.h** tab.

```

/***** WIFI *****/

// the AdafruitIO_WiFi client will work with the following boards:
// - HUZZAH ESP8266 Breakout -> https://www.adafruit.com/products/2471
// - Feather HUZZAH ESP8266 -> https://www.adafruit.com/products/2472
// - Feather M0 WiFi -> https://www.adafruit.com/products/2473
// - Feather WICED -> https://www.adafruit.com/products/3056

#define WIFI_SSID "Test WiFi"
#define WIFI_PASS "my wifi password"

// comment out the following two lines if you are using fona or ethernet
#include "AdafruitIO_WiFi.h"
AdafruitIO_WiFi io(IO_USERNAME, IO_KEY, WIFI_SSID, WIFI_PASS);
  
```

## FONA Config

If you wish to use the FONA 32u4 Feather to connect to Adafruit IO, you will need to first comment out the WiFi support in **config.h**

```
/****** WIFI *****/  
// comment out default  
// wifi config lines  
// - Feather Huzzah ESP8266 -> https://www.adafruit.com/products/2821  
// https://www.adafruit.com/products/3010  
// https://www.adafruit.com/products/3056  
  
#define WIFI_SSID "Test WiFi"  
#define WIFI_PASSWORD "my wifi password"  
  
// comment out the following two lines if you are using fona or ethernet  
// #include "AdafruitIO_WiFi.h"  
// AdafruitIO_WiFi io(IO_USERNAME, IO_KEY, WIFI_SSID, WIFI_PASS);
```

Next, remove the comments from both of the FONA config lines in the FONA section of **config.h** to enable FONA support.

```
/****** FONA *****/  
  
// the AdafruitIO_FONA library:  
// - Feather 32u4 FONA -> https://www.adafruit.com/product/3027  
  
// uncomment the following two lines:  
// comment out the AdafruitIO_WiFi client in the WIFI section  
#include "AdafruitIO_FONA.h"  
AdafruitIO_FONA io(IO_USERNAME, IO_KEY);
```

## Ethernet Config

If you wish to use the Ethernet Wing to connect to Adafruit IO, you will need to first comment out the WiFi support in **config.h**



```

/***** WIFI *****/

// comment out default
// wifi config lines
// - Feather Huzzah ESP8266 -> https://www.adafruit.com/products/2821
// - Feather WiFi -> https://www.adafruit.com/products/3010
// - Feather WiFi -> https://www.adafruit.com/products/3056

#define WIFI_SSID "Test WiFi"
#define WIFI_PASS "my wifi password"

// comment out the following two lines if you are using fona or ethernet
// #include "AdafruitIO_WiFi.h"
// AdafruitIO_WiFi io(IO_USERNAME, IO_KEY, WIFI_SSID, WIFI_PASS);

```

Next, remove the comments from both of the Ethernet config lines in the Ethernet section of **config.h** to enable Ethernet Wing support.

```

/***** ETHERNET *****/

// the Adafruit uncomment both boards:
// - Ethernet FeatherWing -> https://www.adafruit.com/products/3201

// uncomment the
// and comment out the AdafruitIO_WiFi client in the section
#include "AdafruitIO_Ethernet.h"
AdafruitIO_Ethernet io(IO_USERNAME, IO_KEY);

```

Next, we will look at how the example sketch works.

# Code

The **adafruitio\_07\_digital\_out** example uses digital pin 5 by default on all boards, and that can be modified by changing the **LED\_PIN** define.

**Note:** If you are using the WICED Feather, you will need to change the **LED\_PIN** define to *PC5* instead of the default setting of 5.

```
/****** Example Starts Here *****/
```

```
// digital pin 5
#define LED_PIN 5
```

The next chunk of code sets up an Adafruit IO Feed instance for a feed called **digital**.

```
// set up the 'digital' feed
AdafruitIO_Feed *digital = io.feed("digital");
```

In the setup function, we set the **LED\_PIN** as a digital output, and connect to Adafruit IO. We also attach a function called **handleMessage** to the **digital** feed that will be called whenever your device receives messages for that feed.

The code will wait until you have a valid connection to Adafruit IO before continuing with the sketch. If you have any issues connecting, check **config.h** for any typos in your username or key.

```
void setup() {

  // set led pin as a digital output
  pinMode(LED_PIN, OUTPUT);

  // start the serial connection
  Serial.begin(115200);

  // wait for serial monitor to open
  while(! Serial);

  // connect to io.adafruit.com
  Serial.print("Connecting to Adafruit IO");
  io.connect();

  // set up a message handler for the 'digital' feed.
  // the handleMessage function (defined below)
  // will be called whenever a message is
  // received from adafruit io.
```

```
digital->onMessage(handleMessage);

// wait for a connection
while(io.status() < AIO_CONNECTED) {
  Serial.print(".");
  delay(500);
}

// we are connected
Serial.println();
Serial.println(io.statusText());

}
```

Next, we have the main `loop()` function. The first line of the loop function calls `io.run()`; this line will need to be present at the top of your loop in every sketch. It helps keep your device connected to Adafruit IO, and processes any incoming data.

```
void loop() {
  // io.run(); is required for all sketches.
  // it should always be present at the top of your loop
  // function. it keeps the client connected to
  // io.adafruit.com, and processes any incoming data.
  io.run();
}
```

The final chunk of code is the **handleMessage** function. This is the function that is called whenever the **digital** feed gets a message.

We use the `data->toPinLevel()` function to convert the incoming data to either **LOW** or **HIGH**, and set the state of the **LED\_PIN** to that value.

```
// this function is called whenever an 'digital' feed message
// is received from Adafruit IO. it was attached to
// the 'digital' feed in the setup() function above.
void handleMessage(AdafruitIO_Data *data) {

  Serial.print("received <- ");

  if(data->toPinLevel() == HIGH)
    Serial.println("HIGH");
  else
    Serial.println("LOW");

  // write the current state to the led
  digitalWrite(LED_PIN, data->toPinLevel());

}
```

Upload the sketch to your board, and open the Arduino Serial Monitor. Your board should

now connect to Adafruit IO.

Connecting to Adafruit IO....

Adafruit IO connected.

Toggle the button on your Adafruit IO dashboard, and you should see the following in the Arduino Serial Monitor.

```
received <- HIGH  
received <- LOW  
received <- HIGH  
received <- LOW
```

You can now toggle the button on your Adafruit IO dashboard, and you should see your LED turn on and off.